

THE VIRTUAL RESOLUTIONS PACKAGE FOR MACAULAY2

AYAH ALMOUSA, JULIETTE BRUCE, MICHAEL C. LOPER, AND MAHRUD SAYRAFI

ABSTRACT. We introduce the `VirtualResolutions` package for the computer algebra system `Macaulay2`. This package has tools to construct, display, and study virtual resolutions for products of projective spaces. The package also has tools for generating curves in $\mathbb{P}^1 \times \mathbb{P}^2$, providing sources for interesting virtual resolutions.

Recently, Berkesch, Erman, and Smith introduced the notion of virtual resolutions for subvarieties of smooth projective toric varieties as an analogue to minimal graded free resolutions for subvarieties of projective space [BES17]. Broadly, a virtual resolution is a homological representation of a finitely generated graded module over the Cox ring of a smooth projective toric variety.

While graded minimal free resolutions are useful for studying quasicohherent sheaves over projective space, when working over a product of projective spaces or, more generally, over smooth projective toric varieties, they are somewhat less useful. In particular, graded minimal free resolutions over the Cox ring seem too burdened by algebraic structure that is, in some sense, unimportant geometrically. By allowing a limited amount of homology, virtual resolutions offer a more flexible alternative for studying toric subvarieties when compared to graded minimal free resolutions. This article introduces the `VirtualResolutions` package for `Macaulay2` [M2], which provides tools for constructing and studying virtual resolutions over products of projective spaces.

The `VirtualResolutions` project began in 2018 at the *Macaulay2 Workshop at the University of Wisconsin - Madison*, building on previous work of Christine Berkesch, David Eisenbud, Daniel Erman, and Gregory G. Smith. Along with them, the authors of this paper developed the `VirtualResolutions` package to generate examples of virtual resolutions in products of projective spaces. In particular, the package provides tools to generate examples of virtual resolutions and to check if chain complexes are virtual resolutions. These methods are introduced and demonstrated in Section 2.

In Section 3, we concentrate on virtual resolutions arising from curves in $\mathbb{P}^1 \times \mathbb{P}^2$. This is the simplest case after points in a product of projective spaces. We demonstrate several methods for finding the defining ideals of curves in the Cox ring of $\mathbb{P}^1 \times \mathbb{P}^2$, including monomial curves, rational curves, and curves from \mathbb{P}^3 .

ACKNOWLEDGEMENTS

We would like to thank Christine Berkesch, David Eisenbud, Daniel Erman, Gregory G. Smith, and Mike Stillman for helping to develop the `VirtualResolutions` package. Further, we especially thank Christine Berkesch for reading a draft of this article and providing many helpful comments.

This article, and the associated `Macaulay2` package, were started at the *Macaulay2 Workshop at University of Wisconsin - Madison* funded by NSF grant DMS-1812462. Additional work was done during the *Graduate Workshop in Commutative Algebra for Women and Mathematicians of Minority Genders* funded by NSF grant DMS-1908799. The final portion of this article and package was written during a Coding Sprint hosted by the Institute for Mathematics and its Applications (IMA) and we are grateful for their hospitality.

AA was partially supported by the NSF GRFP under Grant No. DGE-1650441.

JB was partially supported by the NSF GRFP under Grant No. DGE-1256259 and NSF grant DMS-1502553.

ML was partially supported by NSF RTG grant DMS-1745638.

2. CONSTRUCTING VIRTUAL RESOLUTIONS

If X is a smooth projective toric variety, we denote its $\text{Pic}(X)$ -graded Cox ring, as defined in [Cox95], by $\text{Cox}(X)$ (see also Section 5.2 of [CLS11]). Denote the associated irrelevant ideal of X by $B \subset \text{Cox}(X)$. A virtual resolution of a $\text{Pic}(X)$ -graded module over $\text{Cox}(X)$ is defined as follows:

Definition 2.1. [BES17, Definition 1.1] A *virtual resolution* of a $\text{Pic}(X)$ -graded $\text{Cox}(X)$ -module M is a chain complex of $\text{Pic}(X)$ -graded free $\text{Cox}(X)$ -modules

$$\mathbf{F}_\bullet := [F_0 \longleftarrow F_1 \longleftarrow F_2 \longleftarrow \cdots]$$

such that the corresponding complex of \mathcal{O}_X -modules is a locally free resolution of the sheaf \widetilde{M} .

Importantly from a computational perspective, the definition of a virtual resolution can be rephrased as follows: a complex \mathbf{F}_\bullet of $\text{Pic}(X)$ -graded free $\text{Cox}(X)$ -modules is a virtual resolution of M if and only if

- (1) The B -saturation of $H_0(\mathbf{F}_\bullet)$ is isomorphic to the B -saturation of M , and
- (2) $H_i(\mathbf{F}_\bullet)$ is supported on B for all $i > 0$.

For the remainder of this paper, we will focus on the case when X is a product of projective spaces, since [BES17] demonstrates several ways to construct virtual resolutions in this case. We let $\mathbf{n} = (n_1, n_2, \dots, n_r) \in (\mathbb{Z}_{\geq 0})^r$ and $\mathbb{P}^{\mathbf{n}} = \mathbb{P}^{n_1} \times \mathbb{P}^{n_2} \times \cdots \times \mathbb{P}^{n_r}$. Further, we write S for $\text{Cox}(\mathbb{P}^{\mathbf{n}})$, which is graded by $\text{Pic}(\mathbb{P}^{\mathbf{n}}) \cong \mathbb{Z}^r$.

Given this set-up, Berkesch, Erman, and Smith show that it is possible to construct a virtual resolution of a \mathbb{Z}^r -graded S -module M from the graded minimal free resolution of M and an element of the multigraded Castelnuovo-Mumford regularity as introduced in [MS04]. More specifically, if $\mathbf{d} \in \mathbb{Z}^r$ is an element of the multigraded Castelnuovo-Mumford regularity of the module M , then Berkesch, Erman, and Smith define the **virtual resolution of the pair** (M, \mathbf{d}) by removing all twists generated in degrees not less than $\mathbf{n} + \mathbf{d}$ from the graded minimal free resolution of M [BES17, Theorem 1.3].

Notice that the virtual resolution of the pair (M, \mathbf{d}) is a subcomplex of the minimal free resolution, so its length is at most the projective dimension of M . Also note that we are justified in saying “the” virtual resolution of the pair (M, \mathbf{d}) because this complex is unique up to isomorphism.

Example 2.2. Consider three points $([1 : 1], [1 : 4])$, $([1 : 2], [1 : 5])$, and $([1 : 3], [1 : 6])$ in $\mathbb{P}^1 \times \mathbb{P}^1$. Using Macaulay2, we may compute the B -saturated ideal defining these three points.

```
i1 : needsPackage "VirtualResolutions";
i2 : X = toricProjectiveSpace(1)**toricProjectiveSpace(1);
i3 : S = ring X; B = ideal X;
o4 : Ideal of S
i5 : J = saturate(intersect(
    ideal(x_1 - x_0, x_3 - 4*x_2),
    ideal(x_1 - 2*x_0, x_3 - 5*x_2),
    ideal(x_1 - 3*x_0, x_3 - 6*x_2)), B);
o5 : Ideal of S
```

One can show the \mathbb{Z}^2 -graded minimal free resolution of S/J is

$$\begin{array}{ccccccc}
& & S(0,-3) & & & & \\
& & \oplus & & & & \\
& & S(-1,-2) & & S(-1,-3)^2 & & \\
& & \oplus & & \oplus & & S(-2,-3) \\
S \longleftarrow & S(-2,-1) & \longleftarrow & S(-2,-2)^2 & \longleftarrow & \oplus & \longleftarrow 0. \\
& \oplus & & \oplus & & S(-3,-2) & \\
& S(-3,0) & & S(-3,-1)^2 & & & \\
& \oplus & & & & & \\
& S(-1,-1) & & & & &
\end{array}$$

We may view the multigraded Betti table for the above graded minimal free resolution.

```

i6 : minres = res J;
i7 : multigraded betti minres
      0      1      2      3
o7 = 0: 1      .      .      .
      2: .      ab      .      .
      3: . a3+a2b+ab2+b3      .      .
      4: .      . 2a3b+2a2b2+2ab3      .
      5: .      .      . a3b2+a2b3
o7 : MultigradedBettiTally

```

In order to compute a virtual resolution of S/J we find an element of the multigraded Castelnuovo-Mumford regularity of the module [MS04]. This can be done via the `multigradedRegularity` command, which utilizes the `TateOnProducts` package to compute the minimal elements in the multigraded regularity of S/J .

```

i8 : multigradedRegularity(X, S^1/J)
o8 = {{0, 2}, {1, 1}, {2, 0}}
o8 : List

```

The function `virtualOfPair` implements Theorem 1.3 in [BES17] to compute the virtual resolution of a pair as defined above. We use it here to compute the virtual resolution of the pair $(S/J, (2, 0))$. Note that since we must remove all twists generated in degrees not less than $\mathbf{n} + \mathbf{d}$ from the graded minimal free resolution, we give `virtualOfPair` the element $(3, 1) = (1, 1) + (2, 0)$.

```

i9 : vres = virtualOfPair(J, {{3, 1}})

```

```

      1      3      2
o9 = S <-- S <-- S <-- 0

```

```

      0      1      2      3
i10 : multigraded betti vres

```

```

      0      1      2
o10 = 0: 1      .      .
      2: .      ab      .
      3: . a3+a2b      .
      4: .      . 2a3b
o10 : MultigradedBettiTally

```

The above virtual resolution of the pair $(S/J, (2, 0))$ is shorter and thinner than the graded minimal free resolution of S/J . We may check that `vres` is indeed a virtual resolution by using the `isVirtual` function.

```

i11 : isVirtual(J,B,vres)
o11 = true

```

By default, `isVirtual` checks whether the annihilator of the homology is supported on the irrelevant ideal. As an alternative approach, we also implement the method for checking whether a complex is virtual presented in Theorem 1.3 of [Lop19]. This is done by setting `Strategy => Determinantal`.

Remark 2.3. One can use the Tate resolution for a product of projective spaces to compute elements of the multigraded regularity [EES15]. The function `multigradedRegularity` included in the `VirtualResolutions` package builds upon the function `cohomologyHashTable` found in the `TateOnProducts` package [EES5] to compute the multigraded Castelnuovo-Mumford regularity of a \mathbb{Z}^r -graded S -module that is B -saturated.

Another way to generate virtual resolutions is by using Theorem 4.1 of [BES17], which provides a way to produce virtual resolutions for ideals defining zero-dimensional subschemes of \mathbb{P}^n . More specifically, if J is a B -saturated ideal defining a zero-dimensional subscheme of \mathbb{P}^n then for some $\mathbf{a} = (a_1, a_2, \dots, a_r) \in (\mathbb{Z}_{\geq 0})^r$ with $a_r = 0$, and other entries sufficiently positive, the minimal free resolution of $S/(J \cap B^{\mathbf{a}})$ is a short virtual resolution of S/J . See Section 4 of [BES17] for the definition of $B^{\mathbf{a}}$. The function `resolveViaFatPoint` computes the minimal free resolution of $S/(J \cap B^{\mathbf{a}})$.

Example 2.4. Again consider the ideal of three points in $\mathbb{P}^1 \times \mathbb{P}^1$ as in Example 2.2. We use the function `resolveViaFatPoint` to compute a virtual resolution of J .

```
i12 : C = resolveViaFatPoint(J,B,{2,0})
```

```

      1      4      3
o12 = S <-- S <-- S <-- 0
      0      1      2      3
```


```
o12 : ChainComplex
i13 : isVirtual(J,B,C)
o13 = true
```

Note that the virtual resolution produced by `resolveViaFatPoint` and the one produced by `virtualOfPair` are different. In fact, virtual resolutions obtained by `resolveViaFatPoint` may not necessarily be obtained using `virtualOfPair`. Thus, `resolveViaFatPoint` is a useful method for producing new and interesting examples of virtual resolutions.

3. CONSTRUCTING CURVES IN $\mathbb{P}^1 \times \mathbb{P}^2$

One source of potentially interesting virtual resolutions comes from studying curves in $\mathbb{P}^1 \times \mathbb{P}^2$. With this in mind, the `VirtualResolutions` package contains functions for constructing a limited class of curves in $\mathbb{P}^1 \times \mathbb{P}^2$. In particular, there are functions for generating monomial curves, rational curves, and curves arising from curves in \mathbb{P}^3 .

The main function in this direction is `curveFromP3toP1P2`. Given the defining ideal of a curve in \mathbb{P}^3 , this function returns the defining ideal of a curve in $\mathbb{P}^1 \times \mathbb{P}^2$ constructed in the following way: given projections $\pi_1 : \mathbb{P}^3 \rightarrow \mathbb{P}^1$ and $\pi_2 : \mathbb{P}^3 \rightarrow \mathbb{P}^2$, the universal property of fibre products gives the diagram



The induced map from \mathbb{P}^3 to $\mathbb{P}^1 \times \mathbb{P}^2$, represented above by the squiggly arrow, is the map under which we are computing the image of our curve C .

The projections that `curveFromP3toP1P2` uses are fixed. In particular, it uses the coordinate projections

$$\pi_1([z_0 : z_1 : z_2 : z_3]) = [z_0 : z_1] \quad \text{and} \quad \pi_2([z_0 : z_1 : z_2 : z_3]) = [z_1 : z_2 : z_3].$$

As one might wish to preserve the degree of the curve $C \subset \mathbb{P}^1 \times \mathbb{P}^2$, the function `curveFromP3toP1P2` has an option called `PreserveDegree`. When this option is set to `true`, the function will return an error if the given curve intersects the base locus of these projections.

Example 3.1. The code below uses `curveFromP3toP1P2` to construct a curve in $\mathbb{P}^1 \times \mathbb{P}^2$ from the defining ideal of the twisted cubic $C \subset \mathbb{P}^3$.

```
i14 : R = ZZ/101[z_0,z_1,z_2,z_3];
i15 : I = ideal(z_0*z_2-z_1^2, z_1*z_3-z_2^2, z_0*z_3-z_1*z_2);
o15 : Ideal of R
i16 : J = curveFromP3toP1P2(I)
      2
o16 = ideal (x      - x      x      , - x      x      + x      x      , - x      x      + x      x      )
          1,1      1,0 1,2      0,1 1,1      0,0 1,2      0,1 1,0      0,0 1,1
          ZZ
o16 : Ideal of ---[x      , x      , x      , x      , x      ]
          101 0,0      0,1      1,0      1,1      1,2
```

We can check that the ideal J defines a curve in $\mathbb{P}^1 \times \mathbb{P}^2$ by computing its dimension. Note that since J is an ideal in the Cox ring of $\mathbb{P}^1 \times \mathbb{P}^2$, we expect J to be three dimensional, since in general the dimension of a subscheme defined by a $\text{Pic}(X)$ -graded ideal $J \subset \text{Cox}(X)$ is given by $\dim J - \text{rank Pic}(X)$.

```
i17 : dim J
o17 = 3
```

Combining `curveFromP3toP1P2` with the `SpaceCurves` package [SSZ,Zha18] allows one to randomly generate curves in $\mathbb{P}^1 \times \mathbb{P}^2$. This is done in the `randomCurveP1P2` function, which uses the `curve` function to generate a curve $C \subset \mathbb{P}^3$ of specified degree and genus. The function `randomCurveP1P2` then constructs a curve $C' \subset \mathbb{P}^1 \times \mathbb{P}^2$ from C using `curveFromP3toP1P2`.

Example 3.2. Using the `randomCurveP1P2` function we produce a random curve in $\mathbb{P}^1 \times \mathbb{P}^2$ coming from a curve of degree 7 and genus 3 in \mathbb{P}^3 . Additionally, we check that the resulting ideal defines a curve.

```
i18 : I = randomCurveP1P2(7,3);
      ZZ
o18 : Ideal of ---[x      , x      , x      , x      , x      ]
          101 0,0      0,1      1,0      1,1      1,2
i19 : S = ring I;
i20 : dim I
o20 = 3
```

Using the `multigradedRegularity` function, one can see that $(2,3)$ is a minimal element in the multigraded regularity of this curve. Note that we first saturate the ideal defining our curve; this is required to ensure that `multigradedRegularity` runs correctly.

```
i21 : B = intersect(ideal(x_(0,0), x_(0,1)), ideal(x_(1,0), x_(1,1), x_(1,2)));
o21 : Ideal of S
i22 : J = saturate(I,B);
o22 : Ideal of S
i23 : multigradedRegularity(S, S^1/J)
o23 = {{1, 4}, {2, 3}}
```

o23 : List

Finally, we can use this element of the multigraded regularity to compute a virtual resolution of S/J .

```
i24 : minres = res J;
i25 : vres = virtualOfPair(J, {{3,5}})
```

```
      1      11      18      8
o25 = S <-- S <-- S <-- S <-- 0
      0      1      2      3      4
```

o25 : ChainComplex

Comparing the multigraded Betti tables of these two resolutions we see that once again the virtual resolution is shorter and less complicated than the graded minimal free resolution.

```
i26 : multigraded betti minres
      0      1      2      3      4
o26 = 0: 1      .      .      .      .
      5: .      a3b2+3a2b3      .      .      .
      6: . a4b2+2a2b4+5ab5      .      .      .
      7: .      .      b7 3a4b3+6a3b4+12a2b5+3ab6      .      .
      8: .      .      .      2ab7 3a4b4+8a3b5+6a2b6      .
      9: .      .      .      .      a2b7 a4b5+3a3b6
```

o26 : MultigradedBettiTally

```
i27 : multigraded betti vres
      0      1      2      3
o27 = 0: 1      .      .      .
      5: . a3b2+3a2b3      .      .
      6: . 2a2b4+5ab5      .      .
      7: .      . 6a3b4+12a2b5      .
      8: .      .      . 8a3b5
```

o27 : MultigradedBettiTally

Remark 3.3. Since the curve generated by `randomCurveP1P2` is, in some sense, random one will often get different virtual resolutions when running the above example. For example, working over a slightly larger finite field than in Example 3.1, we compute the multigraded regularity for 500 curves arising from curves of genus 3 and degree 7 in \mathbb{P}^3 . The resulting distribution of multigraded regularities is shown below.

```
i28 : tally apply(500,i->(
      I := randomCurveP1P2(7,3,ZZ/32003);
      S := ring I;
      B := intersect(ideal(x_(0,0), x_(0,1)),
      ideal(x_(1,0), x_(1,1), x_(1,2)));
      J := saturate(I,B);
      multigradedRegularity(S, S^1/J)
    ))
o28 = Tally{{{1, 4}, {2, 3}} => 167}
      {{{1, 5}} => 170}
      {{{1, 6}} => 163}
```

o28 : Tally

The functions `randomRationalCurve` and `randomMonomialCurve` are alternative methods for generating curves in $\mathbb{P}^1 \times \mathbb{P}^2$. Given two positive integers d and e , these functions construct the defining ideal of the curve arising as the image of the map

$$\mathbb{P}^1 \longrightarrow \mathbb{P}^1 \times \mathbb{P}^2 \quad \text{given by} \quad [t_0 : t_1] \longmapsto ([f_0 : f_1], [g_0 : g_1 : g_2])$$

where the f_i and g_i are forms in $\mathbb{K}[t_0, t_1]$ of degrees d and e , respectively. For `randomMonomialCurve`, these forms are chosen to be monomials, while for `randomRationalCurve` they can be any possible forms of the correct degree.

Example 3.4. Here we use the `randomRationalCurve` function to construct a rational curve of bi-degree $(5, 7)$. Once again we verify that it is in fact a curve by computing its dimension.

```
i29 : I = randomRationalCurve(5,7);
      ZZ
o29 : Ideal of ---[x , x , y , y , y ]
      101 0 1 0 1 2
i30 : dim I
o30 = 3
```

REFERENCES

- [BES17] Christine Berkesch, Daniel Erman, and Gregory G. Smith, *Virtual resolutions for a product of projective spaces* (2017). ArXiv pre-print: <https://arxiv.org/abs/1703.07631>. ↑1, 2, 3, 4
- [Cox95] David A. Cox, *The homogeneous coordinate ring of a toric variety*, J. Algebraic Geom. **4** (1995), no. 1, 17–50. MR1299003 ↑2
- [CLS11] David A. Cox, John B. Little, and Henry K. Schenck, *Toric varieties*, Graduate Studies in Mathematics, vol. 124, American Mathematical Society, Providence, RI, 2011. MR2810322 ↑2
- [EES15] David Eisenbud, Daniel Erman, and Frank-Olaf Schreyer, *Tate resolutions for products of projective spaces*, Acta Math. Vietnam. **40** (2015), no. 1, 5–36, DOI 10.1007/s40306-015-0126-z. MR3331930 ↑4
- [EES] David Eisenbud, Daniel Erman, Frank Schreyer, and Michael E. Stillman, *TateOnProducts: Tate resolutions on products of projective spaces. Version 1.0*. ↑4
- [Lop19] Michael C. Loper, *What Makes a Complex Virtual* (2019). ArXiv pre-print: <https://arxiv.org/abs/1904.05994>. ↑4
- [MS04] Diane Maclagan and Gregory G. Smith, *Multigraded Castelnuovo-Mumford regularity*, J. Reine Angew. Math. **571** (2004), 179–212, DOI 10.1515/crll.2004.040. MR2070149 ↑2, 3
- [M2] Daniel R. Grayson and Michael E. Stillman, *Macaulay 2, a software system for research in algebraic geometry*. Available at <http://www.math.uiuc.edu/Macaulay2/>. ↑1
- [SSZ] Frank Schreyer, Mike Stillman, and Mengyuan Zhang, *SpaceCurves: Generation of space curves. Version 1.0*. ↑5
- [Zha18] Mengyuan Zhang, *The SpaceCurves package in Macaulay2*, J. Softw. Algebra Geom. **8** (2018), 31–48, DOI 10.2140/jsag.2018.8.31. MR3820372 ↑5

DEPARTMENT OF MATHEMATICS, CORNELL UNIVERSITY, ITHACA, NY

E-mail address: aka66@cornell.edu

URL: <http://math.cornell.edu/~aalmousa>

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF WISCONSIN, MADISON, WI

E-mail address: juliette.bruce@math.wisc.edu

URL: <http://math.wisc.edu/~juliettebruce/>

SCHOOL OF MATHEMATICS, UNIVERSITY OF MINNESOTA, MINNEAPOLIS, MINNESOTA, 55455

E-mail address: loper012@umn.edu

URL: <http://math.umn.edu/~loper012/>

SCHOOL OF MATHEMATICS, UNIVERSITY OF MINNESOTA, MINNEAPOLIS, MINNESOTA, 55455

E-mail address: mahrud@umn.edu

URL: <http://math.umn.edu/~mahrud/>